A litany of ladders: easy problems with hard solutions

Alasdair McAndrew
Alasdair.McAndrew@vu.edu.au
College of Engineering and Science
Victoria University
PO Box 14428, Melbourne 8001
Australia

Abstract

Mathematics is full of problems from the easy to the intractable; in fact it may be quite fairly said that mathematics *is* the study of problems. Some problems are straightforward enough to be used as student exercises; others are of a difficulty and complexity to occupy the attention of scholars all their lives. The purpose of this paper is to look at a few problems which are neither trivially easy nor impossibly difficult. These problems all have the similarity that in their classical statements they are about the placement of ladders. However, they are difficult enough that solving them is more than an elementary routine exercise; also, they are perfect vehicles for the use of a Computer Algebra System. We thus show how standard, simple problems can be greatly expanded in scope by the use of technology, and these new problems, which may be seen as "difficult" in a classroom sense, are amenable to experimentation.

1 The crossed ladders problem

This problem has attained a certain degree of notoriety. In his *Bibliography of Recreational Mathematics* [9], William Schaaf says: "Like the cat in the alley, the problem of the crossed ladders seems also to have nine lives." Its basic format can be illustrated as in the left hand diagram in Figure 1.

We imagine an alleyway in which there are two ladders leaning against each wall; the base of each ladder sits firmly at an edge of the alleyway. The lengths of the ladders a and b, and the height of their crossing h is given: what is the width of the alley? This problem was discussed at length by Martin Gardner in one of his "Mathematical Games" columns for Scientific American; it was reprinted as the chapter "Elegant Triangles" in the book "Mathematical Circus" [5]. Schaaf [9] gives references going back to 1909.

To solve this, we introduce some new lengths as shown in the right hand diagram. By similar triangles, we have

$$\frac{y}{h} = \frac{x}{B}, \quad \frac{z}{h} = \frac{x}{A}.$$



Figure 1: The crossed ladders

By adding these equations we have

$$\frac{y+z}{h} = \frac{x}{A} + \frac{x}{B}$$

and since $y+z=x$ we can divide through by x to obtain

$$\frac{1}{h} = \frac{1}{A} + \frac{1}{B}.$$

This is a standard result and is known as the *crossed ladders theorem*. In fact the equation is true as long as the left and right sides, and the line from the crossing, are all parallel—they do not have to be perpendicular to the bottom line. We can rewrite the values A and B in terms of x, a and b by Pythagoras, so that

$$\frac{1}{h} = \frac{1}{\sqrt{a^2-x^2}} + \frac{1}{\sqrt{b^2-x^2}}. \tag{1}$$

This equation can be turned into a polynomial in x, by clearing fractions and by repeated squaring to eliminate all the square roots. This is prone to error, and a much better way is to use a computer algebra system.

To obtain the polynomial, we shall use Sage, and a Gröbner basis reduction to eliminate variables. To do this we can set up a polynomial ring in all the variables, with an ideal generated by the four equations above: the two similar triangle relations, and the relations of the triangle side.

```
sage: R.<a,b,h,x,A,B,y,z> = PolynomialRing(QQ)
sage: Id = R.ideal([A^2+x^2-a^2, B^2+x^2-b^2, y+z-x, y*B-x*h, z*A-x*h])
sage: E = Id.elimination_ideal([A,B,y,z])
sage: p = E.gen(0)
sage: px = list(p.factor())[1][0].polynomial(x)
```

The result of all of this is a lovely eighth degree polynomial, which is in fact a quartic polynomial in x^2:

$$\begin{aligned} &x^8+\left(-2a^2-2b^2+4h^2\right)x^6+\left(a^4+4a^2b^2+b^4-6a^2h^2-6b^2h^2\right)x^4\\ &\quad+\left(-2a^4b^2-2a^2b^4+2a^4h^2+8a^2b^2h^2+2b^4h^2\right)x^2\\ &\quad+a^4b^4-2a^4b^2h^2-2a^2b^4h^2+a^4h^4-2a^2b^2h^4+b^4h^4 \end{aligned} \tag{2}$$

Since quartic polynomials can be solved by radicals, it would be possible, at least in theory, to obtain a closed-form expression for x as a function of a, b, and h. Such an expression however, would be very large, unwieldy, and useless. However the interest of this particular problem is that such an initial simple setting should give rise to such a complicated polynomial to be solved.

What we *can* do, though, is to explore possible integer values of the parameters which give rise to "nice" (integer) values of x. To do this, we can simply loop over values of the parameters—keeping a and b different, and ensuring that h is always less than both of them. For each set of parameters, we obtain the above polynomial and attempt to factorize it. If a factorization is possible, then we obtain a solution for x. For example, with $a = 37$, $b = 91$ and $h = 14$ we find that the polynomial above can be factored into

$$(x^6 - 17291x^4 + 83266003x^2 - 71405259849)(x + 35)(x - 35)$$

whence $x = 35$ seems to be a solution. The following is a brute-force program to find such values:

```
sage: for i in range(1,100):
....:     for j in range(i+1,100):
....:         for k in range(1,min(i,j)):
....:             pp=px.subs(a=i,b=j,h=k,x=w)
....:             pl = pp.factor_list()
....:             if len(pl)>2:
....:                 print i,j,k,pp.factor()
```

With the range given here, the parameters with integer values are

$$\begin{aligned} a,b,h,x &= 37,91,14,35\\ &= 51,75,40,45\\ &= 52,60,45,48\\ &= 75,78,70,72 \end{aligned}$$

However, these results are not in fact solutions to the initial crossed ladders equation (1), since we have left out some scaling factors as part of the simplification when obtaining the eighth degree polynomial above. For example, substituting $a, b, x = 37, 91, 35$ into the right hand side of equation (1) produces

$$\frac{1}{\sqrt{37^2-35^2}}+\frac{1}{\sqrt{91^2-35^2}}=\frac{2}{21}$$

or $h = 21/2$. To obtain an integer value for h, we need to multiply each of a, b, x by 2 to obtain $h = 21$, and so

$$a,b,h,x = 74,182,21,70$$

is a set of integer solutions.

Another set of manageable solutions can be obtained for those values of a, b and h for which the polynomial (2) can be factorized into two quartics; in fact into two quadratics in x^2. For example, if $a, b, h = 20, 28, 15$ then the polynomial factorizes into

$$(x^4 + 16x^2 - 108800)(x^4 - 1484x^2 + 563200)$$

for which the only positive real solution is

$$x = 2\sqrt{18\sqrt{2} - 2} \approx 17.9428.$$

There is in fact a simpler approach, which I traced back to a 1956 paper [2]. Instead of solving for the alley width x, we first solve for the value $A + B$, from which the width x can be obtain by a simple quadratic.

The setup is very similar, except for a new variable $u = A + B$:

```
sage: Id = R.ideal([A^2+x^2-a^2, B^2+x^2-b^2, y+z-x, y*B-x*h, z*A-x*h, u-A-B])
sage: E = Id.elimination_ideal([A,B,x,y,z])
sage: p = E.gen(0)
sage: pu = list(p.factor())[-1][0].polynomial(u)
```

This polynomial is the much simpler

$$u^4 - 4hu^3 - (a^2 - b^2)^2.$$

A similar approach is discussed by Bremner at al [4]; this paper also provides an extensive bibliography.

2 The crossed ladders problem made difficult

The crossed ladders problem, as described in section 1, is difficult enough maybe, involving as it does the solution of a quartic equation in x^2. But in fact it can be attacked in many different ways, some of which we have explored. However, there is one aspect of the problem which can be changed, and that is its physical provenance. We have modelled the ladders by two lines—which as we know may be taken to have zero width. Now modelling the physical world by making simplifying assumptions is standard in mathematics, even if those assumptions are not physically realizable[1]. And it often happens that the more "real world" assumptions we take into account, the more intractable the problem becomes.

So we now change the ladder problem by assuming that the ladders, as well as having lengths a and b, have widths w. We take the same width for each ladder. This just adds *one* more parameter, but as we shall see, explodes the difficulty of the problem. In this new problem, we take h as being the common height of both ladders from the ground.

The new problem is shown on the left in Figure 2, where the ladders are shown in a sort of x-ray view.

The right hand diagram in Figure 2 shows some extra lengths that we will use to develop a system of equations. The shaded part shows two similar triangles.

[1] A classic example of this is in the physics joke about cows: "Consider a spherical cow. . . "



Figure 2: Crossed ladders with non-zero width

By Pythagoras, we immediately have

$$p^2 + q^2 = w^2, \quad r^2 + s^2 = w^2.$$

and by similar triangles,

$$\frac{q}{w} = \frac{y+z+r}{b}, \quad \frac{s}{w} = \frac{y+z+p}{a}, \quad \frac{y}{h} = \frac{q}{p}, \quad \frac{z}{h} = \frac{s}{r}$$

and of course

$$x = p + q + y + z.$$

We can now enter all of this into Sage, and again eliminate variables by a Gröbner basis reduction.

```
sage: R.<a,b,w,x,p,q,r,s,y,z> = PolynomialRing(QQ)
sage: Id = R.ideal([p^2+q^2-w^2,r^2+s^2-w^2,q*b-w*(y+z+r),s*a-w*(y+z+p),\
....: y*p-h*q,z*r-h*s,x-p-r-y-z])
sage: E = Id.elimination_ideal([p,q,r,s,y,z])
sage: px = E.gen(0).polynomial(x)
```

The result is a splendid polynomial in x of degree 12, which is far too large to display. But we can display an example, say with $a, b, w, h = 10, 15, 1, 2$:

```
sage: var('u')
sage: px.subs(a=10,b=15,w=1,h=2,x=u)
```

$$\begin{aligned}&506205001u^{12} - 670470200u^{11} - 322935375884u^{10} + 335225355400u^{9}\\&\quad + 73656660260134u^{8} - 53780411133400u^{7} - 7115882111296900u^{6}\\&\quad + 2781397003913400u^{5} + 258332092075315625u^{4} - 4111492998550000u^{3}\\&\quad - 10743648295954150000u^{2} - 689287839525000000u - 113126876243750000\end{aligned}$$

Sage's `find_root` command will produce $u = 4.105408760589318$ as a positive real root.

We have seen earlier how to find integer values for the standard ladder problem. We now might ask: are there values of a, b, w, x and h which are all simultaneously integers?

We start by considering the two right triangles under any one of the ladders, as shown in Figure 2. Suppose that every length in the larger right hand triangle is a multiple of the lengths of the sides in the smaller left hand triangle. Thus

$$b = mw, \quad y + z + r = mq$$

for some (integer) m, and similarly

$$a = nw, \quad p + y + z = ns$$

for some (integer) n. We need to choose m and n so that

$$x = mq + p = ns + r.$$

From this last equation we can write

$$qm - sn = r - p$$

and by the extended Euclidean algorithm for the greatest common divisor we know that this equation will have a solution in integers if $\gcd(s, q)$ is a factor of $r - p$.

In order to have a non-trivial example, we shall aim for the two small corner triangles: with sides p, q, w and r, s, w, to be different. If they are the same then either the ladder diagram is symmetric, or the ladders meet at right angles. So we first need to find a solution to

$$w^2 = p^2 + q^2 = r^2 + s^2$$

in integers, with $\{p, q\} \neq \{r, s\}$. A little experimentation produces

$$25^2 = 7^2 + 24^2 = 15^2 + 20^2.$$

We thus choose $w, p, q, r, s = 25, 15, 20, 7, 24$. We now have to find integers m and n for which

$$20m + 15 = 24n + 7.$$

This can be rewritten as $24n - 8 = 20m$ and by dividing through by 4 we have

$$2(3n - 1) = 5m$$

and so we can choose n so that $3n-1$ is a multiple of 5, for example $n = 7$. Then $m = 8$. Thus $a = nw = 175$, $b = mw = 200$ and $x = 20m + 15 = 175$. To determine h, we have the ratios of sides of similar triangles:

$$\frac{h}{z} = \frac{7}{24}, \quad \frac{h}{y} = \frac{15}{20}$$

as well as $y + z = x - p - r = 153$. These can be solved as linear equations for h, y, z and we find that

$$h = \frac{3213}{100}$$

We can clear this fraction by multiplying by 100, which leads to the integer values:

$$\begin{aligned} w &= 2500 \\ a &= 17500 \\ b &= 20000 \\ h &= 3213 \\ x &= 17500. \end{aligned}$$

Finally, we display the 12-th degree polynomial with these values. It turns out that the coefficients all have a large common divisor, and once divided out the remaining polynomial is

$$\begin{aligned} &43681\, z^{12} - 147733740\, z^{11} - 60967809507000\, z^{10} + 156933228375000000\, z^{9} \\ &\quad + 3217355851035300000000\, z^{8} - 5568409330682900625000000\, z^{7} \\ &\quad - 7750484271226683593750000000000\, z^{6} + 650862679712990062500000000000000\, z^{5} \\ &\quad + 7792572903944320753051757812500000000000\, z^{4} \\ &\quad + 11329136274273236938476562500000000000000\, z^{3} \\ &\quad - 171553661234895026866149902343750000000000000000\, z^{2} \\ &\quad - 28792033813148365173339843750000000000000000000000\, z \\ &\quad - 124265672846624496436119079589843750000000000000000 \end{aligned}$$

It is clear that this simple generalization of the crossed ladders problem leads to some very complicated mathematics. Here are some questions which arise from the elementary discussion so far:

1. Is there a simple function of the variables (such as $u = A + B$ for the original problem) which leads to a simpler form of the equation?
2. What is the equation with smallest coefficients for which the solution is an integer?
3. By how much more would the problem be complicated if the ladder widths were allowed to be different?
4. What would the equations look like if the cross-sectional areas of the ladders (both assumed to be non-zero) were equal?

The author has experimented with question 1, but has not yet found a solution.

3 The ladder in the corner problem

This ladder problem, problem seems to have attained the status of a modern mathematical classic, at least in terms of elementary calculus, and there must be few students who have not been exposed to it. The problem can be stated as:

> Two hallways of width a and b meet a right angles. What is the longest ladder—assumed to be of zero width—which can be moved around the corner?

Sometimes "ladder" is replaced with "pipe"; at any rate the problem is of moving a long straight object around a corner, as shown in Figure 3.

The insight required to solve this problem is to notice that the *longest* ladder which can be carried around the corner is the length of the *shortest* line between the outer two walls of the hallway, that touches the inner corner. Note that shorn of any physical meaning, and couched in purely abstract terms, the problem asks for the shortest line between the positive x and y axis that passes through the point (a, b), where both coordinates are positive. This is the form in which it is found in old calculus textbooks, such as Hardy's classic *Course of Pure Mathematics* [6].



Figure 3: Moving a zero-width ladder around a corner



Figure 4: Solving the ladder problem

There are two simple methods of solution; one uses trigonometry, the other straightforward algebra, both of which are illustrated in Figure 4. For the trigonometric solution, suppose that the ladder makes an angle θ with the y axis, and that the lengths of the ladder from the corner to the west and south walls are X and Y respectively. Since $X \sin\theta = a$ and $Y \cos\theta = b$ the length of the ladder is then

$$L = \frac{a}{\sin\theta} + \frac{b}{\cos\theta}.$$

Alternatively, suppose that the lengths of the walls reached by the ladder from the outside corners are $a + x$ and $b + y$. By similar triangles, $b/x = y/a$ and so $y = ab/x$. Then we may aim to minimize the square of the length of the ladder, which is

$$L^2 = S = (a + x)^2 + \left(b + \frac{ab}{x}\right)^2.$$

With either equation, finding the stationary points and substituting back produces the minimum length of line, which is the maximum length of ladder, which as we mentioned earlier is $(a^{2/3} + b^{2/3})^{3/2}$.

So far, there is nothing here which is out of the reach of a beginning calculus student. Our intention is to generalize the problem.

4 Ladders with positive width

The problem suddenly rockets into a new sphere of difficulty when we assume that the ladder has a non-zero width, say c, as shown in Figure 5. In this form it is a version of the "piano-mover's problem" which in its most general form involves moving a rigid object between points in the plane.

Applying the trigonometric approach, we see that the length of the original ladder has been reduced by a small amount at each end. Chasing angles and similar triangles reveals that the length of the original ladder has been shortened by $c\cot\theta$ at the top and by $c\tan\theta$ at the bottom. Thus the new length is

$$\frac{a}{\sin\theta} + \frac{b}{\cos\theta} - c\frac{\cos\theta}{\sin\theta} - c\frac{\sin\theta}{\cos\theta}$$

and this can be written more elegantly as

$$\frac{a\cos\theta + b\sin\theta - c}{\sin\theta\cos\theta}.$$



Figure 5: Moving a ladder with positive width

To find the maximum value, again differentiate and solve for θ. Considering the numerator of the derivative only, the equation to be solved is

$$a\cos^3\theta - b\sin^3\theta + c(\sin^2\theta - \cos^2\theta) = 0.$$

This is an intractable looking equation, so we shall make a polynomial of it by applying the transformation $\theta = 2\arctan t$ for which

$$\sin\theta = \frac{2t}{1+t^2}, \quad \cos\theta = \frac{1-t^2}{1+t^2}.$$

Substituting these into the previous equation, and again considering the numerator only, we have the polynomial equation

$$(a+c)t^6 - (3a+5c)t^4 + 8bt^3 + (3a-5c)t^2 - a + c = 0.$$

This is a *sextic equation* for which there is no general solution in terms of radicals: by the well-known Abel-Ruffini theorem. This is the point at which one would ordinarily give up trying to find a closed form solution. We will instead try to find values of the parameters for which this equation can be solved using radicals.

5 Solving polynomial equations

The fundamental theorem of algebra states that all polynomials have a complex root: in fact all polynomials can be factored into linear factors over the complex numbers. For quadratic, cubic and quartic polynomials, the roots can be obtained by a finite sequence of arithmetic operations and the taking of n-th roots. However, as we have known now for 200 years, quintic equations, and equations of higher orders, can not in general be so solved. What this means is that for every $n \geq 5$ there is a polynomial equation of degree n whose roots can *not* be obtained by *any* finite sequence of arithmetic operations and root-taking. Equations whose roots can be obtained by such a sequence of operations are said to be *solvable by radicals*, or simply (by a slight abuse of terminology) *solvable*.

Although the general sextic is unsolvable by radicals, *some* sextics are solvable. Trivially, sextics which are factorizable into a quadratic and a quartic are solvable. More interestingly, though, some irreducible sextics are solvable [3].

Although we might hope to eliminate terms of the sextic, it turns out we will have better luck by factorization. In particular, we will be able to solve the sextic equation (by radicals) if either

- its Galois group is solvable
- it can be factored over $\overline{\mathbb{Q}}$ into polynomials of degree 4 or less, whose coefficients themselves are the roots of solvable equations of degree 4 or less.

Here $\overline{\mathbb{Q}}$ is the algebraic completion of $\mathbb{Q}$: the set of all roots of polynomials over $\mathbb{Q}$. At this stage I'll come clean and admit that I don't know of any method which will enable us to solve the equation in any sort of generality: such a solution, if it existed, would be extremely complicated. Instead, we shall pick some values of a, b and c which should work, and see what we can do with them.

We will need some computer algebra help, and using Sage, we can run through values $300 > a > b > c >= 1$ to find values for which either the sextic is solvable, or is factorizable.

In this range there are no values of a, b, c which gave a factorizable polynomial, and only 11 triples whose polynomial has a solvable Galois group. The smallest such triple is

$$a, b, c = 36, 5, 4$$

but we shall swap the values of a and b, so as to avoid complex numbers later on. We note in passing that although the theory of Galois groups is well understood, effective means of computing them for polynomials of arbitrary size is still an active area of research.

Attempts at factorization

Since the polynomial to be solved is irreducible, we can't factorize over the rationals. Our hope is to be able to factor over the algebraic numbers, and for the coefficients in the factors to be obtainable by solving equations of degree four or less. We shall closely follow the method given by Piezas [8], even down to his choice of symbols.

We can try factorizing into or into two cubics or into a quartic and quadratic. Suppose we try two cubics; we will try to find the coefficients r_i, m, n so that

$$t^6 - \frac{35}{9}t^4 + 32t^3 - \frac{5}{9}t^2 - \frac{1}{9} = (t^3 + r_1t^2 + r_2t + r_3)(t^3 + mt^2 + nt + r_4).$$

First set up the equations.

```
sage: F = expand((t^3+r1*t^2+r2*t+r3)*(t^3+m*t^2+n*t+r4)).collectterms(t)
sage: eqs = [maxima.coeff(F,t,i)-maxima.coeff(L,t,i) for i in range(6)]
```

These equations (which will be automatically set equal to zero) are:

$$r_3 r_4 + \frac{1}{9} = 0 \qquad r_2 r_4 + n r_3 = 0 \qquad r_1 r_4 + m r_3 + n r_2 + \frac{5}{9} = 0$$

$$r_1 + m = 0 \qquad r_4 + r_3 + m r_2 + n r_1 - 32 = 0 \qquad r_2 + m r_1 + n + \frac{35}{9} = 0$$

We can solve the last four equations for r_i:

```
sage: rs = maxima.solve([eqs[2],eqs[3],eqs[4],eqs[5]],[r1,r2,r3,r4])[0]
sage: rs
```

$$\left[r_1 = -m,\ r_2 = -\frac{9n - 9m^2 + 35}{9},\right.$$
$$r_3 = \frac{9n^2 + (9m^2+35)n - 9m^4 + 35m^2 + 288m - 5}{18m},$$
$$\left. r_4 = -\frac{9n^2 + (35 - 27m^2)n + 9m^4 - 35m^2 - 288m - 5}{18m}\right]$$

Now we substitute these into the first two equations:

```
sage: e0 = maxima.subst(rs,eqs[0])
sage: e1 = maxima.subst(rs,eqs[1])
```

What we have now is two very long polynomial equations in m and n. Each polynomial can be considered as a univariate polynomial in one variable, with coefficients being polynomials in the other variable. To eliminate each of m or n we can use the *resultant* of the two polynomials, which is defined as being equal to zero if and only if the original two equations share a common root. In general, given two monic polynomial equations $p(x) = 0$ and $q(x) = 0$ of degrees n and m respectively, their resultant is defined to be

$$\prod_{i=1}^{n}\prod_{j=1}^{m}(p_i - q_j)$$

where p_i and q_j are the roots of p and q. Note that the resultant can in fact be computed without knowing the roots first, as the determinant of a matrix whose rows are shifted lists of the coefficients; this matrix is called *Sylvester's matrix* [1].

```
sage: resn = maxima.resultant(e1,e0,m).factor()
sage: resn
```

$$3842422663603177 4976(n^2 - 10n + 5)(9n^2 + 35n - 5)^2$$
$$(43046721n^{18} + \cdots + 49207488981750n + 20823831125)$$

```
sage: resm = maxima.resultant(e1,e0,n).factor()
sage: resm
```

$$-688747536m^4(9m^2-125)(59049m^{18}-\cdots+4551498748225m^2-1385280125)$$

In this last expression the final term is polynomial in even powers of m only; that is a ninth-degree polynomial in m^2.

Now we can find values of m and n by solving some of the low degree factors in the resultants:

```
sage: n0 = resn.part(2).solve(n)[1].rhs()
sage: m0 = resm.part(1,3).solve(m)[0].rhs()
sage: r40 = maxima.subst(n0,n,maxima.subst(m0,m,rs[3].rhs())).radcan()
sage: m0,n0,r40
```

$$2\sqrt{5}+5,\ -\frac{5^{3/2}}{3},\ -\frac{2\sqrt{5}+5}{3\sqrt{5}}$$

We now have all the coefficients we need for the second cubic factor, and so we can solve this equation:

```
sage: sol = maxima.solve(t^3+m0*t^2+n0*t+r40,t)
sage: sol[2].radcan()
```

$$t=\frac{5^{3/2}}{9}+\frac{\left(729(2^{3/2})\sqrt{901\sqrt{5}+2041}-266(3^{7/2})\sqrt{5}-22(3^{15/2})\right)^{1/3}}{3^{19/6}}-\frac{2^{2/3}(27\sqrt{5}+5)}{3^{11/6}\left(27\sqrt{2}\sqrt{901\sqrt{5}+2041}-133\sqrt{15}-11(3^{9/2})\right)^{1/3}}$$

This has approximate value

```
sage: t0 = sol[2].rhs().float();
sage: t0
```

$$0.26092$$

and can be turned into an angle:

```
sage: t0.atan()*2
sage: (t0.atan()*2*180/pi).float()
```

which is 0.51046 radians or 29.247 degrees.

Note that this accords with the one positive real root of the initial sextic:

```
sage: L.realroots().float()
```

$$[t=-2.8863, t=.26092]$$

A triumph for computer algebra and *really complicated* algebraic expressions. Note that we have suppressed a lot of detail here: for instance, how do we know which factors of the resultants in m and n are to be solved to create r_4? I simply used trial-and-error: trying different values until a pair was found which can be used to provide a root of the original sextic (an approximate value of which can be found by any standard numeric root-finding methods).

Note that in this example the two hallways are of very different widths; we might consider this example as moving something between a large room and a small service corridor, as shown in Figure 6.



Figure 6: The example $(a, b, c) = (5, 36, 4)$

6 More generalization and concluding remarks

In some ways the problem we have just explored is the simplest of all possible generalizations. We could also ask for the longest ladder (of either zero or positive width) which can be moved around a corner where the hallways meet at an angle different from a right angle. Or we could ask, as Leo Moser did in 1966 [7], for the largest area of any shape that can be moved around a corner—and this problem, the so-called "sofa problem" is as yet unsolved.

The problems we have solved and generalized show in part the remarkable power of computer algebra to explore and (attempt to) solve difficult problems. Although none of the mathematics we have discussed is, in itself, conceptually difficult, the complexity of the algebra involved is considerable, and trying to deal with much of this by hand would be an arduous task.

We notice that we have left many problems unsolved (and unmentioned). Here are several:

1. Can we find integer values for all of the parameters a, b, w, x and h?
2. Are there integer values of the parameters for which the 12-th degree equation has a factor of degree 4 or less?
3. Are there integer values for the parameters a, b, c and t of the corner problem?
4. Are there integer values a, b and c which produce a polynomial factorizable over the rationals $\mathbb{Q}$? Or, alternatively, can we prove that there are none?

I have not explored these in great detail; it may be that some of them are relatively straightforward to solve.

I encourage you and your students to roll up your sleeves, crank up your favourite computer algebra system, and enjoy experimenting!

References

[1] Alkiviadis G Akritas. “Sylvester’s forgotten form of the resultant”. In: *Fibonacci Quarterly* 31.4 (1993), pp. 325–332.

[2] H. A. Arnold. “The Crossed Ladders”. In: *Mathematics Magazine* 29 (1956), pp. 153–154.

[3] C. Boswell and M. L. Glasser. *Solvable Sextic Equations*. 2005. arXiv: `math-ph/0504001`.

[4] A Bremner, R Høibakk, and D Lukkassen. “Crossed ladders and Euler’s quartic”. In: *Annales Mathematicae et Informaticae*. Vol. 36. 2009, pp. 29–41. URL: `http://ami.ektf.hu/uploads/papers/finalpdf/AMI_36_from29to41.pdf`.

[5] Martin Gardner. *Mathematical Circus*. New York: Knopf, 1979.

[6] G. H. Hardy. *A Course of Pure Mathematics*. 1st ed. Cambridge University Press, 1908.

[7] Leo Moser. “Problem 66-11: Moving Furniture through a hallway”. In: *SIAM Review* 8 (1966), p. 381.

[8] Titus Piezas III. *Solving Solvable Sextics Using Polynomial Decomposition*. 2004. URL: `http://citeseerx.ist.psu.edu/viewdoc/summary?doi=10.1.1.90.6646`.

[9] William L Schaaf. “A Bibliography of Recreational Mathematics, Volume 2.” In: (1970). URL: `http://files.eric.ed.gov/fulltext/ED040874.pdf`.